
EXECUTIVE SUMMARY

Introduction

Detailed, quantitative interpretations of images require data about the angular distribution of light
reflected from a shadowed sea. This report demonstrates the solution of the two-dimensional
radiative transfer problem for the sea shadowed by an opaque body such as a pier (See Figure 1).
The solution to the problem requires two different theoretical approaches. The solution for the
depth dependence of light radiance is obtained using the theory of Green’s functions (Vladimirov,
1971). The solution to the problem in horizontal direction is obtained with the one-dimensional
Fourier transform method (Morse and Feshbach, 1953).

The algorithm used in this report is based on the analytical solution to the spatially two-dimensional
(Cartesian coordinates x and z) scalar radiative transfer equation in seawater. The 0x axis starts on
the sea surface in the center of the opaque shadow (pier) and increases towards the sun in the
direction orthogonal to the axis of the pier. The 0z axis increases down to the sea bottom.

Because the input parameters did not include the scattering phase function, the algorithm uses
experimental results published in literature. The scattering phase function is estimated through the
regression relationships which are derived by processing all fifteen experimental marine Petzold
phase functions (Haltrin, 1997). This regression relationship couples the scattering phase function
with the scattering coefficient and the single-scattering albedo.

The future possible development of this problem may consists of the following enhancements to
the theory: (a) consideration of the reflection by the sea bottom; (b) consideration of the
atmospheric scattering effects (Haltrin, 1998b, 1998c) and the elevation of the pier; (c)
consideration of the sea surface roughness; (d) generalization to the non-infinite pier (a 3D-problem
in space); (e) consideration of the vertical optical stratification of seawater; and (f) enhancement of
the part of the theory related to the higher-order scattering effects.

Findings and Recommendations

The final equation obtained in this report expresses the upwelling sea radiance distribution as a
function of the sun zenith angle, azimuth and zenith angles of viewing, the width of the pier and
the inherent optical properties of the water (absorption and attenuation coefficients and the
scattering phase function). Results for the actual radiance distribution of light are obtained
numerically with the program RADPIER (see Appendix) using as input the in situ measured
optical properties of seawater. The scattering phase function is estimated through the regression
equations that are based on all fifteen Petzold scattering phase functions. The empirically derived
scattering phase function depends on the scattering coefficient, the single-scattering albedo, and the
scattering angle. The resulting radiance distributions are discussed and illustrated.

INTRODUCTION

Our goal is formulated as follows: Given a downwelling radiance distribution from the sun
consistent with a long rectangular shadow, to model the water-leading radiance as a function of
azimuth and zenith angles of viewing and distance in the direction normal to the long dimension of
the rectangle. The derived approach should be valid for an arbitrary solar elevation.

Several serious attempts have been made to solve two- and three-dimensional radiative transfer
problem (Vladimirov, 1961; Williams, 1967; Kochetkov, 1968; Kaper, 1969; Case and Hazeltine,
1970, Flateau and Stephens, 1988). Unfortunately none of these approaches are simple and
convenient enough to produce practically usable equations applicable to solve our problem.

FORMAL METHODS

We start from the scalar radiative transfer theory (Chandrasekhar, 1960). Our problem is a two
dimensional in space radiative transfer problem. The origin of the Cartesian coordinates is placed
on the sea surface just below the center of an infinite pier. For simplicity we place the pier
horizontally along the 0y axis, and the solar plane is perpendicular to the 0y axis. The 0x axis is
directed horizontally to the sun and lies in the solar plane, and 0z axis is directed vertically to the
bottom.

We use the Green’s function method (Vladimirov, 1971) to solve the problem in a z-direction, and
the Fourier transform method to solve an x-dependent part (Morse and Feshbach, 1953).

The angular distribution of radiance is split into three components: unscattered, single-scattered and
multiple scattered components. Exact solutions for the first two components are obtained.

HOMOGENEOUS SHALLOW SEA ILLUMINATED BY THE SUNLIGHT

We assume that the reader is familiar with the standard terminology of hydrologic optics (see, for
example, Jerlov, 1986; Shifrin 1988, Mobley, 1994).

Let us introduce the following notations: ES  is irradiance by the sun on the sea surface,

E ES S S
0 0= / µ  is irradiance by the sun on the surface normal to the solar rays, µS S SZ h0 = ≡cos sin  is

the cosine of the solar zenith angle ZS , h ZS S= ° −90  is a solar elevation angle.

The solar radiance above the sea surface is expressed by:

L ES S S
0 0 0= −δ µ µ δ ϕ( ) ( ) , (1)

where δ ( )x  is the Dirac’s delta-function (Morse and Feshbach, 1953)

µS S wZ n= −1 2 2sin / (1a)

is the cosine of the light penetration angle, nw  is the seawater refraction coefficient.

If T R ZS F S
↓ ↓= −1 ( ) is a transmission of light by sea surface, where RF  is a Fresnel reflection

coefficient of light falling from above, than the light radiance below the sea surface is:

L Ew w S
0 0( , ) ( ) ( )µ ϕ δ µ µ δ ϕ= − ,   E E Tw S S

0 0= ↓ , (2)

here Ew
0  is the solar irradiance on the surface normal to the rays just below the sea surface.

A one-dimensional scalar radiative transfer equation (Chandrasekhar, 1960) for the total light
radiance can be written as:

µ µ ϕ
π

ϕ γ µ ϕ µ
πd

dz
c L z

b
d p L z d+



 = ′ ′ ′ ′∫ ∫

−

( , , ) (cos ) ( , , )
4

0

2

1

1

, (3)

here c a b= +  is the attenuation coefficient, a  is the absorption coefficient and b  is the scattering
coefficient of seawater, µ θ= cos , the direction of light propagation is determined by the zenith
and azimuth angles θ  and ϕ , p(cos )γ  is the scattering phase function, γ  is the scattering angle
determined from the following formula:

cos cos( )γ µµ µ µ ϕ ϕ= ′ + − − ′ − ′1 12 2 . (4)

Let us represent the total radiance of light inside the sea L as the direct light radiance L0, and the
sum of radiances Ln that represent nth order of scattering:

L b Ln
n

n

=
=

∞

∑
0

. (5)

By substituting Eq. (5) into Eq. (3) we obtain the following equations for radiance components of
different orders of scattering denoted by a subscript:

µ µ ϕd

dz
c L z+



 =0 0( , , ) , (6)

ˆ ( , , ) ˆ ( , , )D L z bS L zn nµ ϕ µ ϕ= −1 , or L z bD S L z bT L zn n n( , , ) ˆ ˆ ( , , ) ˆ ( , , )µ ϕ µ ϕ µ ϕ= =−
− −

1
1 1 . (7)

The operators ̂D  and Ŝ  are defined as follows:

ˆ ( , , ) ( , , ),

ˆ ( , , ) (cos ) ( , , ) ,

D F z
d

dz
c F z

S F z d p F z d

µ ϕ µ µ ϕ

µ ϕ
π

ϕ γ µ ϕ µ
π

≡ +





≡ ′ ′ ′ ′








∫ ∫

−

1
4

0

2

1

1 (8)



Distribution of Light Radiance Reflected from the Sea Illuminated by the Sun with Shadow 9

9

and the following relationships are valid:

ˆ ˆD G− ≡1 ,   ˆ ˆ ˆ ˆ ˆT D S G S= ≡−1 . (8a)

The operator ̂G   in Eq. (8a) is a Green’s function of Eq. (3). This operator is defined according to
the following formula (Vladimirov, 1971):

ˆ ( ) ( ) ( ) ,    ( )
( / ) /G F z G z z F z dz G z

H z
e cz≡ − ′ ′ ′ =

−∞

+∞
−∫ µ

µ
µ , (9)

Here H z( ) is the Heavyside’s or step function H z z H z z( ) ,  ;  ( ) ,  = ≥ = <( )1 0 0 0 .(Morse and
Feshbach, 1953). The total scattering operator is:

ˆ ( , , ) (cos ) ( ) ( , , )T F z d d p dz G z z F zµ ϕ
π

ϕ µ γ µ ϕ
π

≡ ′ ′ ′ − ′ ′ ′ ′∫ ∫ ∫
− −∞

+∞1
4

0

2

1

1

. (10)

Now, the total radiance is expressed as a following sum:

L b Ln
n

n

=
=

∞

∑
0

,   L b T L nn
n n= ≥ˆ ,  0 1, L z b T L zn n

n

( , , ) ˆ ( , , )µ ϕ µ ϕ=
=

∞

∑ 0
0

. (11)

Let us proceed to solve these equations.

SOLUTIONS FOR THE RADIANCE

The solution for direct (unscattered) radiance L0 is straightforwardly obtained from the Eq. (6)

with the following boundary condition on the sea surface, L Lw0
00( , , ) ( , )µ ϕ µ ϕ= :

L z E ew
cz

S
S

0
0( , , ) ( ) ( )/µ ϕ δ µ µ δ ϕµ= −− . (12)

Equation (7) for the single-scattered radiance can be written as:

L z d d p dz G z z L z1

0

2

1

1

0

1
4

( , , ) (cos ) ( ) ( , , )µ ϕ
π

ϕ µ γ µ ϕ
π

≡ ′ ′ ′ − ′ ′ ′ ′∫ ∫ ∫
− −∞

+∞

. (13)

By substituting Eq. (12) into Eq. (13), we have the following solution for the single-scattered
radiance distribution:

L z
E p

c
z H z Hw S

1

0

1 24
( , , )

(cos )
( , ) ( ) ( , ) ( )µ ϕ γ

π µ
ψ µ µ ψ µ µ= + −[ ], (14)

here

ψ µ

µ µ

µ µ

1 1 1( , )
/ /

z
e ecz cz

S

S

= −

−

− −

,   ψ µ µ
1( , ) /z c zeS

cz S= − ,   ψ µ1 0
( , )z c z

z→
= , (15)

ψ µ

µ µ

µ

2 1 1( , )   
/

z
e cz

S

S

=
+

−

. (16)

For the required precision about 15-20% it is sufficient (see Shuleykin, 1968) to calculate the next
term in expansion given by Eq. (11). The explicit solution for the next term can be obtained from
the following equation:

L z d d p dz G z z L z2

0

2

1

1

1

1
4

( , , ) (cos ) ( ) ( , , )µ ϕ
π

ϕ µ γ µ ϕ
π

≡ ′ ′ ′ − ′ ′ ′ ′∫ ∫ ∫
− −∞

+∞

. (17)

with L1  given by Eqs. (14)-(16) and G z( ) given by Eq. (9). The right boundary conditions has no
influence on upwelling radiance distribution, so we do not discuss them here.

SOLUTIONS FOR THE UPWELLING RADIANCE NEAR THE SEA SURFACE

Taking integrals in (17) and summing all required terms at z = 0  and µ < 0, we have the following
solution for the case of homogeneous illumination:

L
E b

c
p

b

c
w S

S
S p( , , ) (cos ) ( , )0

4 2

0

µ ϕ µ
π µ µ

γ ψ µ ϕ=
+( ) +









 ,   µ < 0, (18)

here

cos cosγ µµ µ µ ϕS S S= + − −1 12 2 , (19)

ψ µ ϕ µ
µ µ ϕ

µ µ
µ µ

µ µ
µ µ ϕ µp

p S

S
p

x
d x d( , )

( , , )
( , , )=

′
′ +

′ +
+

′ ′∫ ∫
−0

1

1

0

, (20)

x p p dp S( , , ) (cos ) (cos )µ µ ϕ
π

γ γ ϕ
π

′ = ′ ′∫1
2

0

2

, (21)

cos cos′ = ′ + − ′ − ′γ µ µ µ µ ϕS S S1 12 2 . (22)

HOMOGENEOUS SEA ILLUMINATED BY SUNLIGHT WITH SHADOW

Let us formulate a two-dimensional problem that takes into account an unhomogeneous over
horizontal axis 0x illumination. The two-dimensional radiative transfer equation for the total
angular radiance distribution is:

µ ∂
∂

µ ϕ ∂
∂

µ ϕ
π

ϕ γ µ ϕ µ
π

z x
c L x z

b
d p L x z d+ − +



 = ′ ′ ′ ′∫ ∫

−

1
4

2

0

2

1

1

cos ( , , , ) (cos ) ( , , , ) , (23)

Let us define a one-dimensional Fourier transform by the following equations:

f f x e dxk
ikx= −

−∞

+∞

∫ ( ) ,   f x f e dkk
ikx( ) =

−∞

+∞

∫1
2π

. (24)

After taking a Fourier transform Eq. (23) transfers to the one-dimensional equation for the Fourier
amplitude L zk ( , , )µ ϕ :

µ µ ϕ
π

ϕ γ µ ϕ µ
πd

dz
c L z

b
d p L z dk k+



 = ′ ′ ′ ′∫ ∫

−

˙ ( , , ) (cos ) ( , , )
4

0

2

1

1

, (25)

here

˙ ( )c c ik= +1 τ ,   τ µ ϕ= −1 2 cos c . (26)

Equation (25) may be formally obtained from Eq.(3) by replacing the real extinction coefficient c
by the complex value ċ  given by Eq. (26). It means that the solution of Eq. (24) can be obtained
from the solution (18) by replacing all x-dependent values by their Fourier transforms and all
instances of the extinction coefficient c  by the ̇c . Taking this into account, we have the following
solution to Eq. (25):

L
E b

c
p

b

ck
wk S

S
S p( , , )

˙
(cos )

˙
( , )0

4 2
µ ϕ µ

π µ µ
γ ψ µ ϕ=

+( ) +








 . (27)

Now we only need to calculate a Fourier transform Ewk  of the x-dependent radiance distribution

E E x e dxwk w
ikx= −

−∞

+∞

∫ ( ) . (28)

In our case the solar illumination incorporates a shadowing at − ≤ <w x w , where w  is a half
width of a shadow. The angular-space distribution of the light just under the sea surface is:

L E xw w S
0 ( , ) ( ) ( ) ( )µ ϕ δ µ µ δ ϕ= − , (29)

E x E H w x E w x w xw w w( ) ( ) . sign( ) sign( )= − −[ ] ≡ − − + +[ ]{ }0 01 1 0 5 , (30)

here sign( )x x x= , i.e., sign( ) ,  x x= >1 0 , sign( ) ,  x x= − <1 0.

The Fourier transform of the surface illumination is:

E E k
k

wkwk w= −










0 2
2π δ ( ) sin( ) . (31)

The x-dependent radiance distribution just below the sea surface is:

  L x
E k wk k

ik
p

ik
e dkS w

S
S p

ikx( , , , )
( ) sin( ) ( )

( )
(cos )

( )
( , )0

4 1 2 1
0

0
0µ ϕ µ ω

π µ µ
δ π

τ
γ ω

τ
ψ µ ϕ=

+( )
−

+
+

+










⌠

⌡
−∞

+∞

.(32)

By taking appropriate integrals we have the following solution for the upwelling radiance
distribution just under the sea surface:

L x
E

F w x p F w xS w

S
S p( , , , ) ( , , ) (cos ) ( , , ) ( , )0

4
1

2
10

0

1
0

2µ ϕ ω µ
π µ µ

τ γ ω τ ψ µ ϕ=
+( ) −[ ] + −[ ]


, (33)

F w x w x e w x e

e e
c

w x w x

w x w x

1

2

1
2

1 1

1
2

1

( , , ) sign( ) sign( )

sign ,             
cos

 ,

τ

τ τ µ ϕ

τ τ

τ τ

= + −






+ − −



















+ −






=

−

− + − −

− − − +
(34)

F w x F w x w x e w x e

w x e w x e

w x w x

w x w x

2 1

1
2

1
2

( , , ) ( , , )

( ) ( ) ,

τ τ
τ

τ

τ τ

τ τ

= + − − +












− − + +












− − − +

− − − +
. (35)

Far from the shadow in each direction (at w x w x− + >>,  τ ) F1 0→  and F2 0→  and Eq. (33)
coincides with the equation for upwelling radiance (18) for homogeneous illumination.

Equations (33)-(40) include functions ψ p  and xp, given by Eqs. (20)-(21), that involve
integrations over angular variables µ  and ϕ . In order to simplify these expressions let us
substitute phase functions inside integrals in Eqs (20)-(21) by their transport equivalents:

p B B(cos ) ( ) ( ) ( )γ δ µ µ δ ϕ ϕ→ + − − ′ − ′2 2 1 2 , (36)

p B BS S S S(cos ) ( ) ( ) ( ),   ′ → + − ′ − ′ − =γ δ µ µ δ ϕ ϕ ϕ2 2 1 2 0 , (37)

B p d=
−∫0 5

1

0
. ( )µ µ . (38)

In this case after appropriate integrations we have:

x B p B pp S( , , ) ( , ) ( ) (cos )µ µ ϕ µ µ γ′ = ′ + −2 2 1 2 , (39)

p p d( , ) (cos )µ µ
π

γ ϕ
π

′ = ′∫1
2

0

2

, (40)

ψ µ ϕ µ µ
µ

µ µ
µ µp

S

S

B B
B B B

( , ) ( ) log
( ) ( )

= −
+

+
− + −[ ]

+
4 1 2

1 4 1 1 2 2

. (41)

RADIANCE ABOVE THE SEA SURFACE

In order to calculate radiance distribution above the sea surface as a function of viewing angles, we
have to do the following: a) we should take into account the transmission by sea-air interface from
below by multiplying result by the transmission coefficient:

T RS F
↑ ↑= −( ) ( )µ µ1 ; (42)

b) we should express cosine µ  through the zenith viewing angle θ :

µ θ= −1 2 2sin / nw ; (43)

It means that we should make the following two substitutions in Eq. (33):

E E T Z T E
R Z R

Zw S S S S S

F S F

S

0 0
1 1

→ ≡
−[ ] −[ ]↓ ↑

↓ ↑

( ) ( )
( ) ( )

cos
µ

θ
, (44)

and

µ θ→ −1 2 2sin / nw ; (45)

here RF
↓ and RF

↑ are Fresnel reflection coefficients of sea surface for the light falling from above
and below:

R
n

n

n

n

n

nF
w w

w w

w w

w w
w

w

w

↓ = −
+







+ −
+



















=
− +

( ) ,  µ µ µ
µ µ

µ µ
µ µ

µ
µ1

2

1
2 2 2 2

, (46)

R
n

n

n

n
nF

w a

w a

w a

w a
a w

↑ = −
+







+ −
+



















= −1
2

1
2 2

2 2cos
cos

cos
cos

,  sin
θ µ
θ µ

θ µ
θ µ

µ θ . (47)

APPROXIMATION FOR THE PHASE FUNCTION OF SCATTERING

The backscattering coefficient of light in seawater is connected with the average cosine of light
distribution by the following equations (Haltrin, 1985, 1998a; Haltrin and Kattawar, 1993):

b
a x

xB =
−1

,   x   
( )= −
+ −
1

1 4

2 2

2 4

µ
µ µ

, (48)

This regression is derived by the author from the experimental results by Timofeyeva (1971):

µ

ω

= + − + + − +

+ + − + = − ≡

y y y y

y y y y b c

( . ( . ( . ( .

 ( . ( . . ))))),  ,

2 6178398 4 6024180 9 0040600 14 59994

14 83909 8 117954 1 8593222 1 0

(49)

An empirical representation of the Petzold’s experimental angular scattering coefficient β θ( ) and of
the scattering phase function p( )θ , where θ   is the scattering angle in degrees, are represented by
the following equations (Haltrin, 1997):

β θ θ( ) exp ( )= + −
















=
∑q kn

n

n

n

1 1 2

1

5

, (50)

q b b b b= + − +2 598 17 748 16 722 5 932. . . . , (51)

k

k

k

k

k

1 0

2 0

3 0

4 0

5 0

1 188 0 688

0 1 3 07 1 90

0 01 4 58 3 02

0 001 3 24 2 25

0 0001 0 84 0 61

= −
= −( )

= −( )
= −( )

= −( )














. . ,  

. . . ,

. . . ,

. . . ,

. . . ,

ω
ω
ω
ω
ω

 (52)

p
b

q kn
n

n

n

( ) exp ( )θ π θ= + −
















=
∑4

1 1 2

1

5

(53)

The strong regressions given by Eqs. (50)-(53) can be used as a basis for the empirical model of
the phase functions with the coefficients dependent on the absorption and scattering coefficients.
The single-scattering albedo used here varies from 0.09 to 0.96.

EXAMPLES OF COMPUTATIONS

To illustrate presented theory and algorithm, several examples of the sea shadowed by a long pier
are presented here. All calculations are made with the FORTRAN program RADPIER listed in
Appendix A. Below we present one line graph (Fig. 2) that illustrates the range of changes in the
reflected radiance inflicted by the pier and six density graphs (Figs.3-8) that illustrate the two-
dimensional pictures of reflected radiances.

The one-dimensional graph presented by Fig. 2. shows radiance distribution of reflected light over
0x axis in the direction perpendicular to the pier. The values of inherent optical properties are taken
as follows; c m= −0 5 1. , b m= −0 2 1. ; the solar zenith angle is taken to be 45°, and the solar azimuth
angle is equal to 0° (the solar plane is orthogonal to the pier direction); the viewing zenith angle is
set to 30°. Here we considered two viewing cases: (1) viewing azimuth angle equal to 15°, and (2)
viewing azimuth angle is equal to 105°. In Fig. 2 we see a very sharp shadow with a peculiar
spatial behavior that is a result of backscattering from the sea. Zero values of radiance in this
picture are results of neglecting the part of illumination that is due to the diffuse light of the sky.

Next two figures (Fig. 3 and Fig. 4) show two-dimensional density plots of normalized sea
radiance as a function of coordinate x and azimuth angle of viewing. Zenith angle of viewing is
30°, solar zenith angle is 45°. Fig. 3 displays the radiance with progressive gray-scale palette and
Fig. 4 displays the same values as Fig. 3 only plotted with the banded gray-scale palette to show a
fine structure of the light field.

The next two figures (Fig. 5 and Fig. 6) display two-dimensional density plot of normalized sea
radiance as a function of coordinate x and azimuth angle of viewing. Zenith angle of viewing is
35°, solar zenith angle is 45°. Fig. 5 displays the radiance with progressive gray-scale palette and
Fig. 6 displays the same values as Fig. 5 only plotted with the banded gray-scale palette to show
the fine structure of the light field.

The last two figures (Fig. 7 and Fig. 8) are similar to the previous figures only for different angles:
zenith angle of viewing is 75°, and solar zenith angle equal to 45°. The difference in figures 7 and 8
are also in palettes: Fig. 7 is drawn with progressive gray-scale palette, and Fig. 8 is drawn with
the banded gray-scale palette to show a fine structure of the light field.

CONCLUSIONS

The equations (44)-(47) expresses the upwelling sea radiance distribution as a function of solar
zenith angle, azimuth and zenith angles of viewing, the width of the pier and the inherent optical
properties of the water (absorption and attenuation coefficients and the phase function of
scattering). Results for the actual radiance distribution of light are obtained numerically using the
in situ measured optical properties of seawater and are shown in Figs. 3 – 8. The scattering phase
function used in these calculation is estimated through the regression equations (50)–(53) that are
derived from an analysis of all fifteen Petzold scattering phase functions. The derived scattering
phase function depends the scattering coefficient, the single-scattering albedo, and the scattering
angle.

The main final result of this report consists of the Eqs. (33)-(35) for the upwelling sea angular
radiance distribution as a function of: 1) the solar zenith angle, 2) azimuth and zenith angles of
viewing, 3) the width of the opaque body, 3) absorption and attenuation coefficients, and 4) the
phase function of scattering.



Distribution of Light Radiance Reflected from the Sea Illuminated by the Sun with Shadow 16

16

    

0

0.1

0.2

0.3

0.4

0.5

0.6

0.7

0.8

0.9

1

1.1

1.2

-10 -8 -6 -4 -2 0 2 4 6 8 10

R
at

io
 o

f 
ra

di
an

ce
 w

it
h 

th
e 

sh
ad

ow
 t

o 
th

e 
ba

ck
go

un
d 

ra
di

an
ce

Coordinate x orthogonal to the pier, in m

Case 1

Case 2

Sun zenith angle = 45°
Sun azimuthal angle = 0°

c = 0.5 1/m
b = 0.2 1/m

Water optical properties:

Pier (width = 3 m)

Case 1:
Viewing zenith angle = 30°
Viewing azimuth angle = 15°

Case 2:
Viewing zenith angle = 30°
Viewing azimuth angle = 105°



Distribution of Light Radiance Reflected from the Sea Illuminated by the Sun with Shadow 17

17

Fig. 3. Two-dimensional density plot of normalized sea radiance as a function of coordinate x and
azimuth angle of viewing. Zenith angle of viewing is 30°, solar zenith angle is 45°.

Fig. 4. Two-dimensional density plot of normalized sea radiance as a function of coordinate x and
azimuth angle of viewing. Zenith angle of viewing is 30°, solar zenith angle is 45°. The
same as Fig. 3 only plotted with the banded gray-scale palette to show the fine structure of
the light field.



Distribution of Light Radiance Reflected from the Sea Illuminated by the Sun with Shadow 19

19

Fig. 5. Two-dimensional density plot of normalized sea radiance as a function of coordinate x and
the azimuth angle of viewing. Zenith angle of viewing is 35°, solar zenith angle is 45°.

Fig. 6. Two-dimensional density plot of normalized sea radiance as a function of coordinate x and
azimuth angle of viewing. Zenith angle of viewing is 35°, solar zenith angle is 45°. The
same as Fig. 5 only plotted with the banded gray-scale palette to show the fine structure of
the light field.

Fig. 7. Two-dimensional density plot of normalized sea radiance as a function of coordinate x and
azimuth angle of viewing. Zenith angle of viewing is 75°, solar zenith angle is 45°.

Fig. 8. Two-dimensional density plot of normalized sea radiance as a function of coordinate x and
azimuth angle of viewing. Zenith angle of viewing is 75°, solar zenith angle is 45°. The
same as Fig. 7 only plotted with the banded gray-scale palette to show the fine structure of
the light field.

Appendix A

FORTRAN PROGRAM RADPIER

c **********************************************************************
c                          program RadPier
c **********************************************************************
c         Created May 5, 1997. Last modification: May 9, 1997
c            Written by Vladimir I. Haltrin, NRL Code 7331
c        Ph.: (228) 688-4528, e-mail: <haltrin@nrlssc.navy.mil>
c **********************************************************************
c                    This code can be downloaded
c     from an anonymous ftp site at <serv7330.nrlssc.navy.mil>
c **********************************************************************
c     This program calculates an x-profile of radiance reflected by the
c     homogeneous sea illuminated by the direct sun light and shadowed
c     by the pier of 2w width (in 0x direction) and infinitely stretched
c     in the 0y direction. The depth is measured along the 0z axis.
c     The radiance L is a function of two viewing angles thet and fi,
c     and horizontal coordinate x (which is perpendicular to the pier).
c     L is also parametrically dependent on the inherent optical
c     parameters of water (b, c), the sun zenith angle zang, and the
c     halfwidth of the pier w.
c **********************************************************************
c       This program is based on the two-dimentional in space (x, z)
c     theoretical algorithm derived in April 1997 by V.I. Haltrin
c     specifically for the cooperative task No. 5979-A7 with the NRL
c     Code 7240 (now Code 7340) (P.I. Dr. Peter M. Smith).
c       The algorithm is based on the scalar radiative transfer theory.
c     It uses Green's function approach to solve the problem in
c     z-direction and the Fourier-transform approach to solve the
c     problem in x-direction. In some integrals in order to receive
c     simple analytical solutions the phase-functions are approximated
c     by the transport phase function.
c       It is possible to rewrite this code without such approximation;
c     but in that case it will incorporate two-dimentional numerical
c     integrations.
c **********************************************************************
      implicit   none
      integer    i,j,M,N,Nx,Na
      parameter  (N=501,M=181)
      real*8     x(N),L(N,M),fia(M)
      real*8     a,c,w,zang,thet,fi,x0,dx
      real*8     b,omeg,bp,bb,mua,mus,Rb,R0,theta,mu,xx
      real*8     psip,fsip,angd,fourpi,phf,mux,fis
      real*8     F1,F2,tau,op,L0,rL11,xi,dfi
      real*8     muav,muwa,xgor,fpsip,cosgam,pphf,ftau ! functions
      character  tb
      data       fourpi /12.5663706/

      open(11, file='radpier.in', status='old')
         read(11,*) b ! water scattering coefficient in 1/m
         read(11,*) c ! water attenuation coefficient in 1/m
         read(11,*) w ! pier halfwidth in m
         read(11,*) zang ! sun zenith angle in degrees
         read(11,*) thet ! viewing zenith angle in degrees
         read(11,*) Na ! Number of viewing azimuth angles in degrees
         read(11,*) Nx ! size of x-array (Nx ≤ N)
         read(11,*) x0 ! initial x coordinate in m
         read(11,*) dx ! spatial step in positive 0x direction in m
      close(11)

      a = c-b ! absorption coefficient
      omeg = b/c ! single-scattering albedo
      mua = muav(omeg) ! average cosine
      mus = muwa(zang) ! cosine of the sun zenith angle in water
      mu =  muwa(thet) ! cosine of zenith viewing angle in water
      xx = xgor(mua) ! Gordon's coefficient x=bb/(a+bb)
      bb = a*xx/(1.-xx) ! backscattering coefficient
      bp = bb/b ! backscattering probability
      psip = fpsip(bp,mu,mus) ! auxilary function of the theory
      L0 = mus*omeg/fourpi
      L0 = L0/(mus+mu)
      op = 0.5*omeg*psip
      fis = 0. ! the sun plane is orthogonal to the pier
      dfi = 180./(Na-1) ! increment in viewing azimuth angle
      do j = 1, Na
         fi = dfi*(j-1)
         fia(j) = fi    ! viewing azimuth angle
         mux = cosgam(mu,mus,fi,fis)   ! cos of direct sun rays in water
         angd = ACOSD(mux)
         phf = pphf(b,omeg,angd)    ! scattering phase function
         tau = ftau(c,mu,fi)
         do i = 1, Nx
            xi   = x0+dx*(i-1)
            x(i) = xi
            call ff12(w,xi,tau, F1,F2)
            L(i,j) = L0*((1.-F1)*phf+(1.-F2)*op) ! Total radiance
         end do
      end do

      rL11 = 1./L(1,1)
      tb = CHAR(9)
      open(21, file='radpier.inf', status='new')
         write(21,55) 'Distribution of radiance reflected by the sea'
         write(21,55) 'shadowed by long pier oriented along axis 0y.'
         write(21,55) '---------------------------------------------'
         write(21,66) 'water scattering coefficient in 1/m,    b=',b
         write(21,66) 'water attenuation coefficient in 1/m,   c=',c
         write(21,66) 'pier halfwidth in meters,               w=', w
         write(21,66) 'sun zenith angle in degrees,         zang=',zang
         write(21,66) 'viewing zenith angle in degrees,     thet=',thet
         write(21,66) 'Function        phf = pphf(b,omeg,angd) = ',phf
         write(21,66) 'Function             op = 0.5*omeg*psip = ',op
         write(21,66) 'initial x coordinate in meters,        x0=',x0
         write(21,66) 'spatial step in positive 0x direction, dx=',dx
      close(21)

      open(22, file='radpier.out', status='new',recl=1800)
         write(22,73) Nx,tb,Na
         write(22,73) 0.,tb,0.
         write(22,75)   x(1),(tb, x(i), i=2,Nx)
         write(22,77) fia(1),(tb,fia(j),j=2,Na)
         do i = 1, Nx
            write(22,88) rL11*L(i,1),(tb,rL11*L(i,j),j=2,Na)
         end do
      close(22)

   55 format(a45)
   66 format(a42,f8.3)
   73 format(i4,a1,i4)
   75 format(f8.2,200(a1,f8.2))
   77 format(f8.2,180(a1,f8.2))
   88 format(f8.6,180(a1,f8.6))

      end

c **********************************************************************
      real*8 function muav(omega)
c **********************************************************************
c    This regression is derived by the author (V.I.Haltrin) from the
c    experimental results by V. A. Timofeyeva, “Optical characteristics
c    of turbid media of the seawater type,” Izv., Atm. Ocean. Physics,
c    Vol.7, No.12, pp.1326-1329 (English translation: pp.863-865), 1971.
c **********************************************************************
      implicit   none
      real*8     omega, y

      y = SQRT(1.-omega)
      muav = y*(2.6178398+y*(-4.6024180+y*(9.0040600+
     &       y*(-14.59994+y*(14.83909+y*(-8.117954+1.8593222*y))))))

      return
      end

c **********************************************************************
      real*8 function muwa(thet)
c **********************************************************************
c     Here a text-book formula is coded.
c **********************************************************************
      implicit   none
      real*8     thet,z,nw
      data       nw /1.34/

      z = SIND(thet)/nw
      muwa = SQRT(1.-z*z)

      return
      end

c **********************************************************************
      real*8 function xgor(mua)
c **********************************************************************
c     Formula used here is taken from the book chapter by V.I. Haltrin:
c     “Propagation of Light in a Sea Depth,” Ch. 2 in: Remote Sensing of
c     the Sea and the Influence of the Atmosphere (in Russian), Moscow-
c     Berlin-Sevastopol, Publ. by the GDR Academy of Sciences Institute
c     for Space Research, pp. 20-62, 1985.
c **********************************************************************
      implicit   none
      real*8     mua,mu2,d

      mu2 = mua*mua
      d = 1.-mu2
      d = d*d
      d = d/(1.+mu2*(4.-mu2))
      xgor = d

      return
      end

c **********************************************************************
      real*8 function fpsip(bp,mu,mus)
c **********************************************************************
c     Formula used here is derived by V.I. Haltrin specifically for the
c     task No. 5979-A7 for the NRL code 7240 (P.I. Peter Smith)
c **********************************************************************
      implicit   none
      real*8     bp,mu,mus,lm,fp,ps,dm

      lm = LOG(1.+ 1./mu)
      lm = mu*lm
      fp = 1.-bp
      dm = fp-bp
      ps = bp*fp*lm
      ps = ps+(bp*fp*mus+dm*dm*mu)/(mus+mu)
      fpsip = 4.*ps

      return
      end

c **********************************************************************
      real*8 function pphf(b,omeg,angd)
c **********************************************************************
c        The equations for this function are taken from: V. I. Haltrin,
c     “Theoretical and empirical phase-functions for Monte-Carlo calcu-
c     lations of light scattering in sea water.” in: Proceedings of the
c     Fourth International Conference  Remote Sensing for Marine and
c     Coastal Environments: Technology and Applications,” Vol. I,
c     Publication by Environmental Research Institute of Michigan,
c     Ann Arbor, Michigan, USA, pp. 509-518, 1997.
c        These equations approximate all 15 Petzold phase functions
c     and relate them with the scattering coefficient b and single-
c     scattering albedo omeg; angd is the scattering angle in degrees.
c **********************************************************************

      implicit   none
      real*8     b,omeg,angd,sa,sb,q
      real*8     k1,k2,k3,k4,k5,sm,bt,pi
      data       pi /3.14159265/

      sa = SQRT(angd)
      sb = SQRT(b)
      q  = 2.598+ sb*(17.748+sb*(-16.722+5.932*sb))
      k1 = 1.188-0.688*omeg
      k2 = 0.1*(3.07-1.9*omeg)
      k3 = 0.01*(4.58-3.02*omeg)
      k4 = 0.001*(3.24-2.25*omeg)
      k5 = 0.0001*(0.84-0.61*omeg)
      sm = sa*(-k1+sa*(k2+sa*(-k3+sa*(k4-k5*sa))))
      bt = EXP(q*(1.+sm))
      pphf = (4.*pi*bt)/b

      return
      end

c **********************************************************************
      real*8 function cosgam(mu,mup,fi,fip)
c **********************************************************************
c     Here a text-book formula is coded.
c **********************************************************************
      implicit   none
      real*8     mu,mup,fi,fip,cg

      cg = COSD(fi-fip)
      cg = cg*SQRT(1.-mup*mup)
      cg = cg*SQRT(1.-mu*mu)
      cg = cg+mu*mup
      cosgam = cg

      return
      end

c **********************************************************************
      real*8 function ftau(c,mu,fi)
c **********************************************************************
c        This is auxilary function for the subroutine ff12
c **********************************************************************
      implicit   none
      real*8     c,mu,fi,tau

       tau = SQRT(1.-mu*mu)
       tau = tau*COSD(fi)
       tau = tau/c
       if (tau.eq.0) tau = 1.e-99
      ftau = tau

      return
      end

c **********************************************************************
      subroutine ff12(w,x,tau, f1,f2)
c **********************************************************************
c     Formulae used here is derived by V.I. Haltrin specifically for
c     the task No. 5979-A7 for the NRL Code 7240 (P.I. Pit Smith)
c **********************************************************************
      implicit   none
      real*8     w,x,tau,f1,f2, w1,w2,s1,s2,st,e1,e2,a1,a2,at,d1,d2

      w1 = w-x
      w2 = w+x
      s1 = 1.
      if (w1.le.0.)  s1 = -1.
      s2 = 1.
      if (w2.le.0.)  s2 = -1.
      st = 1.
      if (tau.le.0.) st = -1.
      e1 = ABS(w1/tau)
      e1 = EXP(-e1)
      e2 = ABS(w2/tau)
      e2 = EXP(-e2)
      d1 = s1*(1.-e1)+s2*(1-e2)+st*(e2-e1)
      f1 = 0.5*d1
      a1 = ABS(w1)
      a2 = ABS(w2)
      d1 = (a1*e1-a2*e2)/tau
      at = ABS(tau)
      d2 = (w1*e1+w2*e2)/at
      f2 = f1 + 0.5*(d1-d2)

      return
      end
c **********************************************************************

A sample of an input file 'radpier.in' (starts just after the dotted line):

------------------------------------------------------------------------
 0.2 b = water scattering coefficient in 1/m
 0.5 c = water attenuation coefficient in 1/m
 1.5 w = pier halfwidth in meters
 45. zang = sun zenith angle in degrees
 25. thet = viewing zenith angle in degrees
 181 Na = number of viewing azimuth angles in degrees (O- π)
 201 Nx = size of x-array (Nx ≤ N)
 -10. x0 = initial x coordinate in meters
 0.1 dx = spatial step in the positive 0x direction in m

